

## 综合案例二 图书管理系统

### 【实验目的与要求】

1. 熟练使用 VFP 的操作环境
2. 熟练使用项目管理器管理各种文件
3. 掌握数据库的设计方法与步骤
4. 熟练掌握数据库表及其表之间关系的创建与使用
5. 掌握程序设计的基本方法
6. 掌握表单及表单中各种控件的设计与使用
7. 掌握类的创建和使用
8. 掌握菜单的设计与使用
9. 掌握应用程序创建和使用
10. 要求连编后的应用程序能正确运行
11. 要求学生独立完成，完成后请上传到 ftp 服务器上

### 【系统功能模块】




图 1 系统功能模块图

### 【模块功能分析】

1. 系统登录：通过密码验证后进入图书管理系统主界面，否则不能运行图书管理系统，以确保系统的安全性。
2. 数据信息管理：主要是图书信息管理，包括记录的浏览。
3. 借阅管理：主要实现图书的借出和归还管理。
4. 统计查询：统计过期末还图书的情况，以及师生人数。
5. 打印输出：实现读者借阅明细的打印输出。
6. 退出系统：退出图书管理系统。

### 【数据库表结构】

表 1 读者表 (dz.dbf)

字段名	数据类型	宽度	字段含义	其他
dzbh	字符型	8	读者编号	主索引
xm	字符型	8	姓名	
lx	字符型	4	类型	字段注释：要么为教师，要么为学生
xb	字符型	2	性别	字段有效性规则：性别只能为男或女！
jg	字符型	10	籍贯	普通索引
yxmc	字符型	20	院系名称	普通索引
csrj	日期型	8	出生日期	
gszt	逻辑型	1	挂失状态	默认值为.F.
zp	通用型	4	照片	

表 2 图书表 (ts.dbf)

字段名	数据类型	宽度	字段含义	其他
sh	字符型	20	书号	主索引
sm	字符型	50	书名	
cbs	字符型	30	出版社	普通索引
zz	字符型	40	作者	
dj	数值型	5, 1	单价	
rkcs	数值型	2, 0	入库册数	记录的有效性规则：入库册数大于等于库存册数
kccs	数值型	2, 0	库存册数	两个字段的“匹配字段类型到类”均为 spinner
bz	备注型	4	备注	

表 3 借阅表 (jy.dbf)

字段名	数据类型	宽度	字段含义	其他
dzbh	字符型	8	读者编号	普通索引
sh	字符型	20	书号	普通索引
jsrq	日期型	8	借书日期	字段有效性规则：借书日期不能比系统日期大
hsrq	日期型	8	还书日期	

### 【实验内容】

1. 在 D 盘新建文件夹 TSGL，复制素材到 TSGL 文件夹中，启动 VFP，设置默认的工作文件夹，例如，set default to d:\tsgl。

2. 打开的项目文件 tsgl.pjx, 素材中已提供了项目、数据库、表、视图, 数据库表已设置了字段属性、表属性以及索引等。

3. 对照图 2 建立 tsk 数据库中表与表之间的关系。





图 2 数据库表之间的关系

#### 4. 创建查询

利用查询设计器创建查询, 文件名存为 chaxun1, 各界面截图如下:



查询设计器截图——输出项



查询设计器截图——联接条件



查询设计器截图——排序依据

### 5. 建立类库文件 mylib.vcx

应用程序的很多界面都要用到“退出”命令按钮，为了提高效率，在类设计器中自定义命令按钮类 cmdexit，并保存在类库文件 mylib.vcx 中，如图 3 所示。按钮中显示的图片来自素材中的 exit.ico 文件，设置命令按钮的 click 事件代码为：thisform.release。




图 3 用户自定义类创建界面

## 6. 创建表单

### i. 系统登录界面 (login.scx)



图 4 用户登录界面

系统登录表单能够实现管理员的登录功能，输入用户名、密码，用户名、密码正确就进入图书管理系统菜单，否则直接结束。设计步骤如下：

(1) 新建一表单以文件名 login.scx 保存。表单居中显示，caption 属性为“登录”。

(2) 按照图 4 所示绘制三个标签和两个文本框，设置标签的相关属性，并拖动到适当位置，设置 text2 的 passwordchar 属性值为“#”。

(3) 添加一个命令按钮，caption 为“确定”，并设置 picture 属性为 ok.ico 图片文件。由用户自定义类 cmdexit 创建“退出”按钮（以后用到的“退出”按钮同样方法添加）。

(4) “确定”按钮的 click 事件代码为：

```
yhm=thisform.text1.value
mm=thisform.text2.value
if yhm="tsgl" and mm="123456"
    do tsglmenu.mpr    &&由于菜单在后面设计，调试时本行可用 messagebox () 替代。
    thisform.release
endif
if yhm<>"tsgl"
    =messagebox("用户名错，请重新输入！")
    thisform.text1.value=""
    thisform.text1.setfocus
endif
if mm<>"123456"
    =messagebox("密码错，请重新输入！")
    thisform.text2.value=""
    thisform.text2.setfocus
endif
```

(5) “退出”按钮由自定义命令按钮类 cmdexit 生成。



ii. 系统引导界面 (start.scx)



图 5 系统引导封面

系统引导封面（如图 5）是运行系统后出现的欢迎界面，显示 10 秒后进入系统登录界面。设计步骤如下：

(1) 新建表单并以文件名 start.scx 保存。设置表单的 autocenter 属性为.T，picture 属性为 fm.jpg 图片文件，titlebar 属性为“0-关闭”。在表单的 init 事件中添加代码：

```
public n      &&定义全局变量 n
n=0
```

(2) 在适当位置添加三个标签控件，caption 属性分别为“欢迎进入”、“图书馆信息管理系统”和“按任意键继续……”。标签的 backstyle 属性为“0-透明”。设置标签的字体、字号和字型及颜色属性。

(3) 添加一个时钟控件 timer1，设置 interval 属性值为 1000，timer 事件代码为：

```
n=n+1
if n=10
do form login.scx  &&10 秒后自动运行登录表单 login.scx
thisform.release
endif
```

(4) 在表单的 keypress 事件中编写代码：

```
if nkeycode=13  &&判断是否按下回车键，10 内按回车键也出现登录界面
do form login.scx
thisform.release
endif
```

iii. 图书信息管理界面 (tsgl.scx)



图 6 图书信息管理表单设计界面

图书信息管理界面主要用于对 ts 中图书记录信息浏览、添加、修改以及删除操作，如图 6 所示。设计步骤如下：

(1) 新建一表单并保存为 tsgl.scx。表单的标题为“图书信息管理”，居中显示，表单的 height 为 400，width 为 600。表单数据环境中添加 ts 表。

(2) 从数据环境中将 ts 表中字段拖动到表单形成相应控件，适当调整位置，形成图 6 所示格局，将 ts 表从数据环境中拖动到表单形成一个表格控件 grdts，设置表格的 deletemark 属性为.F.，readonly 属性为.T.，设置表格的 height 为 200，width 为 400。

(3) 参考图 6，在表单上添加一形状控件绘制一个矩形框，并设置 specialeffect 属性为“0-3 维”，并置于其他控件的下层，再添加两个标签，caption 分别为“图书信息”和“(可直接在图书信息处修改记录)”，按照图示拖放到适当位置。

(4) 新建一个命令按钮组，利用生成器产生 6 个命令按钮，标题分别为“第一条”、“上一条”、“下一条”、“最后一条”、“增加”和“删除”，垂直放置在适当位置。

(5) 命令按钮组的 click 事件代码为：

```
thisform.grdts.column1.text1.setfocus  &&焦点设置在表格当前记录的第一列上
do case
  case this.value=1
    go top
  case this.value=2
    if not bof()
      skip -1
    endif
  case this.value=3
    if not eof()
      skip
    endif
  case this.value=4
    go bottom
  case this.value=5
    if this.command5.caption="增加"
      select ts
      append blank
      thisform.refresh
      thisform.txtsh.setfocus
      thisform.commandgroup1.command5.caption="保存"
      &&单击“增加”按钮后该按钮标题即为“保存”，提示添加的记录需要保存
    else
      x=messagebox("确定增加该记录吗？",1+32,"提示")
      if x=1
        messagebox("记录被保存！")
        thisform.commandgroup1.command5.caption="增加"
      else
        delete
        pack
        go top
```

```

thisform.commandgroup1.command5.caption="增加"
thisform.grdts.recordsource="ts"
for i=1 to thisform.grdts.columncount
    thisform.grdts.columns(i).width=62  &&设置表格各列的宽度
endfor
endif
endif
case this.value=6
y=messagebox("确定删除该记录吗?",1+48,"提示")
if y=1
r=recno()
delete
pack
go r-1  &&跳转到上一条记录
thisform.grdts.recordsource="ts"
for i=1 to thisform.grdts.columncount
    thisform.grdts.columns(i).width=62  &&设置每一列等宽
endfor
endif
endcase
thisform.refresh

```

(6) 设置表格控件的 afterrowcolchange 事件代码为:

```

select ts
this.setall("dynamicforecolor","iif(this.column1.text1.value=ts.sh,;
rgb(255,0,0),rgb(0,0,0))","column")  &&设置表格的当前选定记录为红色显示
thisform.refresh

```

(7) 图 7 为完成后运行的效果图。



图 7 图书信息管理表单运行效果



iv. 借书管理界面 (jsgl.scx)



图 8 借书管理表单设计界面

借书管理表单的功能是将选定的图书和读者信息记录到 jy 表中。如果库存册数不为 0 和读者没有挂失，则可以借出，否则就不允许借出。具体操作步骤如下：

(1) 新建一表单文件 jsgl.scx，参考图 8 所示，设置表单的标题为“借书管理”，表单居中显示，在表单的数据环境中添加 ts、jy 和 dz 三张数据表。

(2) 添加一标签控件，caption 属性为“请输入或选择图书书号：”，添加一组合框控件 combol，coloumcount 属性为 2，设置 rowsourcetype 为“6-字段”，rowsource 为“ts.sh,sm”。

(3) 设置 combol 的 interactivechange 事件代码为：

```
select ts
Thisform.refresh
```

(4) 在组合框 combol 的后面添加一命令按钮 command1，设置 caption 为“确定 (\<0)”，default 为.T.，command1 的 click 事件代码为：

```
select ts
varsh=thisform.combol.text    &&text 属性返回组合框中的值
count to m for sh=allt(varsh)  &&count 命令用于统计记录数
if m=0
    messagebox("您输入的书号有误，请重输！",64,"提示")
else
    locate for sh=allt(varsh)
    thisform.refresh
endif
```

(5) 从数据环境中将 ts 表中字段拖动到表单形成相应的控件，依图 8 所示放置到适当的位置，再添加一形状控件，设置 specialeffect 属性为“0-3 维”，并设置该形状控件覆盖在 ts 表各字段控件的上面（避免运行的时候通过鼠标获得文本框焦点，修改 ts 表中记录值）。

(6) 添加一选项按钮组控件 optiongroup1, 具有两个选项, 标题分别为“浏览图书信息”和“浏览读者信息”, 设置选项按钮组的背景色为淡蓝色。

(7) 设置 optiongroup1 的 click 事件代码为:

```
do case
  case this.value=1
    select ts
  case this.value=2
    select dz
endcase
```

(8) 添加一个具有四个命令按钮的命令按钮组用来浏览当前工作区中表中的记录。参考图书信息管理表单, 编写命令按钮组的 click 事件。

(9) 添加另一个标签控件, caption 属性为“请输入或选择读者编号:”, 添加组合框控件 combo2, coloumcount 属性为 2, 设置 rowsourcetype 为“6-字段”, rowsource 为“dz. dzbh, xm”。

(10) 设置 combo2 的 interactivechange 事件代码为:

```
select dz
thisform.refresh
```

(11) 在组合框 combo2 的后面添加命令按钮 command2, 设置 caption 为“确定 (\<K)”, 设置该按钮的 click 事件代码参照步骤 4 命令按钮 command1 的 click 代码完成。

(12) 设置 combo2 的 keypress 事件代码为:

```
if nkeycode=13
  thisform.command2.click  &&调用 command2 的 click 事件代码
endif
```

(13) 从数据环境中将 dz 表中相应字段拖动到表单形成相应的控件, 按照图 8 所示放置到适当的位置, 并设置照片的 olbzp 控件的 stretch 属性为“2-变比填充”, 再添加一形状控件, 设置 specialeffect 属性为“0-3 维”, 并设置该形状控件覆盖在 dz 表各字段控件的上面(避免运行的时候修改 dz 表中记录值)。

(14) 添加一命令按钮控件, 按照图 8 所示设置 caption 为“借出当前图书”, picture 属性设置为相应的图片文件, 设置 click 事件代码为:

```
select ts
if kccs>0          &&该图书库存不为 0, 可以借出
  select dz
  if gszt=.T.      &&处于挂失状态的读者无法借书
    messagebox("该读者已经挂失, 无法借书!")
  else
    varsh=allt(thisform.txtsh.value)    &&取出当前图书号
    vardzbh=allt(thisform.txtzbh.value)  &&取出当前读者编号
    insert into jy(dzbh, sh, jsrq) values(vardzbh, varsh, date())
    &&将借阅信息追加到借阅表
    select ts
    replace kccs with kccs-1             &&图书表相应图书 kccs 减 1
    thisform.refresh
  endif
else
  messagebox("库存数为零, 借书失败!")
endif
```

(15) 图 9 为完成后运行效果图之一。



图 9 借书管理表单运行效果

v. 还书管理界面 (hsgl.scx)




图 10 还书管理表单设计界面

还书管理表单实现读者的还书功能。选定或者输入读者的信息，表格显示该读者的所有借阅信息，选定表格中还没有归还的图书，实现还书的功能。具体操作步骤如下：

(1) 新建一表单文件 hsgl.scx，如图 10 所示，设置表单标题为“还书管理”，表单居中显示，宽度为 570，高度为 430，在表单的数据环境中添加 ts、jy 和 dz 三张数据表。

(2) 添加一标签控件，caption 为“请输入读者编号：”，添加文本框 text1，在 text1 后添加一命令按钮，caption 为“确定”，default 属性为.T.，设置该按钮的 click 事件代码为：

```
select dz
set order to dzbh &&设置索引 dzbh 为主控索引
seek allt(thisform.text1.value)
```

```

if not found()      &&没有找到相应读者信息
    messagebox("读者编号错误, 请重输!", 0, "提示")
    thisform.text1.value=""
    thisform.text1.setfocus
else                &&找到该读者, 再查看 jy 表, 看是否有借书记录
    thisform.refresh
    select jy
    locate for dzbh=allt(thisform.text1.value)
    if eof()
        messagebox("没有借书记录!")
    else
        thisform.grid1.recordsourcetype=4
        thisform.grid1.recordsource="select ts.sh as 书号,ts.sm as 书名,:
        jy.jsrq as 借书日期,hsrq as 还书日期 from ts,jy ;  &&分号前有空格
        where jy.sh=ts.sh and jy.dzbh=allt(thisform.text1.value) ;
        into cursor jytmp"
        thisform.grid1.column1.width=90
        thisform.grid1.column2.width=200
        thisform.grid1.column3.width=100
        thisform.refresh
    endif
endif
endif
select dz
set order to  &&取消主控索引

```

(3) 按照读者信息管理界面的设置方法, 从数据环境中将 dz 表的字段拖动上了形成相应控件, 将性别文本框改成一个具有两个选项的选项按钮组, 并水平放置, 设置 controlsource 属性为 dz.xb 字段, 再添加一个形状控件设置为三维效果覆盖在读者信息各个控件的上面。

(4) 添加一个标签控件, caption 属性为“读者信息浏览:”, 并在该标签的下面绘制一个具有四个按钮的浏览表中记录的命令按钮组, 设置命令按钮组的 click 事件代码为:

```

select dz
do case
    case this.value=1
        go top
    case this.value=2
        if not bof()
            skip -1
        endif
    case this.value=3
        if not eof()
            skip
        endif
    case this.value=4
        go bottom
endcase

```



```

thisform.refresh
thisform.text1.value=thisform.txtdbh.value
thisform.grid1.recordsourcetype=4
thisform.grid1.recordsource="select ts.sh as 书号,ts.sm as 书名,
jy.jsrq as 借书日期,jy.hsrq as 还书日期 from ts,jy :  &&分号前有空格
where jy.sh=ts.sh and jy.dzbh=allt(thisform.text1.value) :
into cursor jytmp"
thisform.grid1.column1.width=90
thisform.grid1.column2.width=200
thisform.grid1.column3.width=100
thisform.refresh

```

(5) 添加一个表格控件 grid1, 设置表格控件的 readonly 属性为 .T., deletemark 属性为 .F., 表格的宽度为 528, 设置表格控件的 init 事件代码为:

```

this.recordsourcetype=4
this.recordsource="select jy.dzbh as 读者编号,ts.sh as 书号,
ts.sm as 书名,jy.jsrq as 借书日期 from ts,jy :  &&分号前有空格
where ts.sh=jy.sh into cursor jytmp"  &&表格初始数据显示所有借阅信息
this.column1.width=60
this.column2.width=90
this.column3.width=155

```

(6) 添加一命令按钮, caption 属性为“还选择的图书”, picturce 属性设置为相应的图片文件, 设置还书命令按钮的 click 事件代码为:

```

vardzbh=allt(thisform.txtdbh.value)
varsh=jytmp.书号  &&jytmp 为表格的数据源
select jy
locate for dzbh=vardzbh and sh=varsh
if not eof()
  if hsrq<>{}
    messagebox("该书已归还!",0,"提示")
    return .t.
  else
    replace hsrq with date()  &&设置还书日期为当前日期
    select ts
    set order to sh
    seek varsh
    replace kecs with kecs+1  &&还书成功, 库存册数加 1
    thisform.grid1.recordsourcetype=4
    thisform.grid1.recordsource="select ts.sh as 书号,
    ts.sm as 书名,jy.jsrq as 借书日期,jy.hsrq as 还书日期 :
    from ts,jy where jy.sh=ts.sh and jy.dzbh=vardzbh :  &&分号前有空格
    into cursor jytmp"
    thisform.grid1.column1.width=90
    thisform.grid1.column2.width=200
    thisform.grid1.column3.width=100

```



```

        thisform.refresh
    endif
else
    messagebox("没有找到相应的借书记录!", 0, "提示")
endif

```

(7) 图 11 就是运行还书表单后的效果图。



图 11 还书管理表单运行效果

#### vi. 打印读者借阅明细界面 (dzjymxdy.scx)



图 12 打印输出设计界面

该表单可以实现打印输出读者借阅情况。具体设计步骤如下：

(1) 新建一表单，保存为 dzjymxdy.scx。caption 属性为“打印读者借阅明细界面”，居中显示。

(2) 参照图 12，在表单上添加一标签控件，caption 属性为“请输入读者编号:”，autosize 为.T.，添加一文本框 text1。

(3) 在 form1 的 destroy 事件中编写代码：

```
close tables all
```

(4) 参考图示在表单上添加一个命令按钮，caption 属性为“打印”，picture 属性设置为相应的图片文件，default 属性为.T.，在命令按钮的 click 事件中编写代码：

```

public vardzbh
vardzbh=thisform.text1.value
report form dzjymx preview

```

(5) 添加“退出”命令按钮。

## 7、主菜单设置

在项目管理器中创建一般菜单，保存为 mainmenu.mnx 并生成同名的.mpr 菜单程序文件。表

**8 mainmenu.mnx 菜单结构**

菜单名称	结果	执行命令或过程	选项设置
数据信息管理(\<D)	子菜单		快捷键: CTRL+D
图书信息管理	命令	Do form tsgl	
借阅管理(\<Y)	子菜单		快捷键: CTRL+Y
借书管理	命令	Do form jsjl	
-			
还书管理	命令	Do form hsgl	
统计查询(\<T)	子菜单		快捷键: CTRL+T
过期未还图书 查询(\<G)	命令	do chaxun1.qpr	快捷键: CTRL+G
-			
统计读者表中 教师和学生的 人数(\<R)	命令	select lx as 类型, count(*) as 人数 from dz group by 1	快捷键: CTRL+R
打印读者借阅明细 (\<M)	命令	do form dzjymxdy	快捷键: CTRL+M
退出(\<Q)	过程	close all clear events set sysmenu to defa	快捷键: CTRL+Q

## 8、创建主程序

主程序文件是程序的起始执行点，可以是项目中的一个程序文件，也可以是一个表单或者一个菜单。一个项目只可以设置一个主文件，且用粗体显示。

创建一个主程序 main.prg，并设置为主文件，程序代码如下：

```

Modify windows screen title "图书管理系统"      &&设置主窗口的标题
Zoom window screen max                        &&使得主窗口最大化
Do form start.scx
Read events

```

## 9、应用程序的连编及运行

在构造好主程序后，设置主程序 main.prg 主文件，连编项目以编译应用程序，生成可执行的 tsgl.exe 应用程序文件。

